



VLOOKUP function

Chapter 8

Content

- Technical details
- How to get start
- Combine data from several tables onto one worksheet by using VLOOKUP
- Common problem
- Best practices

Use VLOOKUP when you need to find things in a table or a range by row. For example, look up a price of an automotive part by the part number, or find an employee name based on their employee ID.

In its simplest form, the VLOOKUP function says:

=VLOOKUP(What you want to look up, where you want to look for it, the column number in the range containing the value to return, return an Approximate or Exact match – indicated as 1/TRUE, or 0/FALSE).

Source: <https://support.microsoft.com/en-us/excel>

Technical details



Technical details

Use the VLOOKUP function to look up a value in a table.

Syntax

VLOOKUP (lookup_value, table_array, col_index_num, [range_lookup])

For example:

- =VLOOKUP(A2,A10:C20,2,TRUE)
- =VLOOKUP("Fontana",B2:E7,2,FALSE)
- =VLOOKUP(A2,'Client Details'!A:F,3,FALSE)

Source: <https://support.microsoft.com/en-us/excel>

Argument name	Description
lookup_value (required)	<p>The value you want to look up. The value you want to look up must be in the first column of the range of cells you specify in the table_array argument.</p> <p>For example, if table_array spans cells B2:D7, then your lookup_value must be in column B.</p> <p>Lookup_value can be a value or a reference to a cell.</p>
table_array (required)	<p>The range of cells in which the VLOOKUP will search for the lookup_value and the return value. You can use a named range or a table, and you can use names in the argument instead of cell references.</p> <p>The first column in the cell range must contain the lookup_value. The cell range also needs to include the return value you want to find.</p>

Source: <https://support.microsoft.com/en-us/excel>

col_index_num (required)	The column number (starting with 1 for the left-most column of table_array) that contains the return value.
range_lookup (optional)	A logical value that specifies whether you want VLOOKUP to find an approximate or an exact match: <ul style="list-style-type: none">▪ Approximate match - 1/TRUE assumes the first column in the table is sorted either numerically or alphabetically, and will then search for the closest value. This is the default method if you don't specify one. For example, =VLOOKUP(90,A1:B100,2,TRUE).▪ Exact match - 0/FALSE searches for the exact value in the first column. For example, =VLOOKUP("Smith",A1:B100,2,FALSE).

Source: <https://support.microsoft.com/en-us/excel>

How to get start



How to get started

There are four pieces of information that you will need in order to build the VLOOKUP syntax:

1. The value you want to look up, also called the lookup value.
2. The range where the lookup value is located. Remember that the lookup value should always be in the first column in the range for VLOOKUP to work correctly. For example, if your lookup value is in cell C2 then your range should start with C.
3. The column number in the range that contains the return value. For example, if you specify B2:D11 as the range, you should count B as the first column, C as the second, and so on.
4. Optionally, you can specify TRUE if you want an approximate match or FALSE if you want an exact match of the return value. If you don't specify anything, the default value will always be TRUE or approximate match.

Now put all of the above together as follows:

=VLOOKUP(lookup value, range containing the lookup value, the column number in the range containing the return value, Approximate match (TRUE) or Exact match (FALSE)).

Source: <https://support.microsoft.com/en-us/excel>

Examples

Here are a few examples of VLOOKUP:

Example 1

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(B3,B2:E7,2,FALSE)			
11	Result	Olivier			
12					

VLOOKUP looks for *Fontana* in the first column (column B) in table_array B2:E7, and returns *Olivier* from the second column (column C) of the table_array. FALSE returns an exact match.

Source:
<https://support.microsoft.com/en-us/excel>

Example 2

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(102,A2:C7,2,FALSE)			
11	Result	Fontana			

VLOOKUP looks for an exact match (FALSE) of the last name for 102 (lookup_value) in the second column (column B) in the A2:C7 range, and returns *Fontana*.

Example 3

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=IF(VLOOKUP(103,A1:E7,2,FALSE)="Sousa","Located","Not found")			
11	Result	Not found			

IF checks to see if VLOOKUP returns *Sousa* as the last name of employee corresponding to 103 (lookup_value) in A1:E7 (table_array). Because the last name corresponding to 103 is *Leal*, the IF condition is false, and *Not found* is displayed.

Example 4

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=INT(YEARFRAC(DATE(2014,6,30), VLOOKUP(105,A2:E7,5, FALSE), 1))			
11	Result	59			
12					
13					
14					
15					

VLOOKUP looks for the birth date of the employee corresponding to 105 (lookup_value) in the A2:E7 range (table_array), and returns 03/04/1955. Then, YEARFRAC subtracts this birth date from 2014/6/30 and returns a value, which is then converted by INT to the integer 59.

Source:
<https://support.microsoft.com/en-us/excel>

Example 5

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63

8					
9					

10	Formula	=IF(ISNA(VLOOKUP(105,A2:E7,2,FALSE)) = TRUE, "Employee not found", VLOOKUP(105,A2:E7,2,FALSE))			
----	----------------	--	--	--	--

11	Result	Burke			
----	---------------	-------	--	--	--

IF checks to see if VLOOKUP returns a value for last name from column B for 105 (lookup_value). If VLOOKUP finds a last name, then IF will display the last name, otherwise IF returns *Employee not found*. ISNA makes sure that if VLOOKUP returns #N/A, then the error is replaced by *Employee not found*, instead of #N/A.

In this example, the return value is *Burke*, which is the last name corresponding to 105.

Source:
<https://support.microsoft.com/en-us/excel>

**Combine data from several
tables onto one worksheet
by using VLOOKUP**




You can use VLOOKUP to combine multiple tables into one, as long as one of the tables has fields in common with all the others. This can be especially useful if you need to share a workbook with people who have older versions of Excel that don't support data features with multiple tables as data sources - by combining the sources into one table and changing the data feature's data source to the new table, the data feature can be used in older Excel versions (provided the data feature itself is supported by the older version).

	A	B	C	D	E	F	G	H	I	J	K	L
1	Client Code	Attorney	Contact Dat	Time	Time O	Total Hos	Bill Ra	Total Bill	Client Name	Billing Address	Billing City	Billing Sta
2	1179	Noble, Rosanna	01/03/19	1:06 PM	1:09 PM	0.05	\$325	\$16.61	File, Grant	9541 SE Addison St.	Michigan City	IN
3	5246	Frazier, Kris	01/03/19	2:34 PM	5:21 PM	2.79	\$550	\$1,532.33	Harris, Corinne	246 Acacia Ave.	Ellenwood	GA
4	5807	Noble, Rosanna	01/04/19	10:06 AM	3:43 PM	5.61	\$325	\$1,823.21	Carey, Heather	98 Galvin Rd.	Canyon County	CA
5	7632	Shelton, Kasey	01/05/19	8:03 AM	6:48 PM	10.74	\$625	\$6,712.38	Snook, Anthony	7 Blackburn Dr.	South Lyon	MI
6	9238	Shelton, Kasey	01/05/19	1:48 PM	6:28 PM	4.67	\$625	\$2,920.08	Carbajal, Scotty	396 Griffin Drive	Pickerington	OH
7	6327	George, Kellie	01/05/19	2:19 PM	6:10 PM	3.85	\$625	\$2,405.61	Huntington, Jan	973 Tallwood Dr.	Benton Harbor	MI
8	5523	Shelton, Kasey	01/06/19	9:43 AM	12:09 PM	2.42	\$625	\$1,513.88	Ruster, Randall	71 East State St.	Williamport	PA
9	3321	Key, Vicky	01/06/19	11:39 AM	3:02 PM	3.38	\$400	\$1,353.76	Bradley, Kristi	768 Greenrose Ave.	Ambler	PA
10	2758	Boatright, Roosevelt	01/06/19	12:01 PM	12:41 PM	0.67	\$400	\$268.24	Erickson, Tara	312 Devonshire St.	Springfield	PA
11	9575	George, Kellie	01/06/19	6:07 PM	6:46 PM	0.64	\$625	\$402.27	Brown, Charity	7730 Peachtree Rd.	Apopka	FL
12	1179	Noble, Rosanna	01/07/19	8:29 AM	9:11 AM	0.70	\$325	\$228.23	File, Grant	9541 SE Addison St.	Michigan City	IN
13	3321	Key, Vicky	01/07/19	9:24 AM	6:30 PM	9.09	\$400	\$3,634.82	Bradley, Kristi	768 Greenrose Ave.	Ambler	PA
14	9575	George, Kellie	01/07/19	11:21 AM	7:03 PM	7.69	\$625	\$4,806.66	Brown, Charity	7730 Peachtree Rd.	Apopka	FL
15	3713	Noble, Rosanna	01/07/19	11:56 AM	1:40 PM	1.73	\$325	\$563.28	Hume, Merle	42 Campfire Drive	Mebane	NC
16	9400	Frazier, Kris	01/07/19	6:01 PM	6:54 PM	0.88	\$550	\$483.49	Horn, Frances	77 Thompson Drive	Clarksburg	WV
17	1790	Flores, Chasity	01/08/19	1:29 PM	6:49 PM	5.34	\$475	\$2,535.33	Totten, Carlos	3 Fairground Dr.	Mount Holly	NC
18	5246	Frazier, Kris	01/09/19	9:55 AM	5:52 PM	7.95	\$550	\$4,371.50	Harris, Corinne	246 Acacia Ave.	Ellenwood	GA
19	4156	Meyer, Lela	01/10/19	11:52 AM	12:50 PM	0.97	\$550	\$534.76	McDaniel, Evelyn	326 Lookout Drive	Fuquay Varina	NC
20	2006	Boatright, Roosevelt	01/10/19	5:42 PM	7:58 PM	2.26	\$400	\$902.35	Bartels, Fredric	78 Jones Drive	Upland	CA
21	6327	George, Kellie	01/10/19	7:34 PM	7:40 PM	0.10	\$625	\$62.21	Huntington, Jan	973 Tallwood Dr.	Benton Harbor	MI
22	8590	Flores, Chasity	01/13/19	1:18 PM	3:20 PM	2.03	\$475	\$964.59	Harding, Marcella	163 Logan St.	Egg Harbor Township	NJ
23	9142	Key, Vicky	01/13/19	6:19 PM	7:34 PM	1.24	\$400	\$496.06	Pruitt, Barbara	599 Wayne Road	Duluth	GA

Here, columns A-F and H have values or formulas that only use values on the worksheet, and the rest of the columns use VLOOKUP and the values of column A (Client Code) and column B (Attorney) to get data from other tables.

Source: <https://support.microsoft.com/en-us/excel>

1. Copy the table that has the common fields onto a new worksheet, and give it a name.
2. Click **Data > Data Tools > Relationships** to open the Manage Relationships dialog box.



Source: <https://support.microsoft.com/en-us/excel>

3. For each listed relationship, note the following:

- The field that links the tables (listed in parentheses in the dialog box). This is the **lookup_value** for your VLOOKUP formula.
- The Related Lookup Table name. This is the **table_array** in your VLOOKUP formula.
- The field (column) in the Related Lookup Table that has the data you want in your new column. This information is not shown in the Manage Relationships dialog - you'll have to look at the Related Lookup Table to see which field you want to retrieve. You want to note the column number (A=1) - this is the **col_index_num** in your formula.

4. To add a field to the new table, enter your VLOOKUP formula in the first empty column using the information you gathered in step 3.

In our example, column G uses Attorney (the **lookup_value**) to get the Bill Rate data from the fourth column (**col_index_num** = 4) from the Attorneys worksheet table, tblAttorneys (the **table_array**), with the formula **=VLOOKUP([@Attorney],tbl_Attorneys,4,FALSE)**.

The formula could also use a cell reference and a range reference. In our example, it would be **=VLOOKUP(A2,'Attorneys'!A:D,4,FALSE)**.

5. Continue adding fields until you have all the fields that you need. If you are trying to prepare a workbook containing data features that use multiple tables, change the data source of the data feature to the new table.

Source: <https://support.microsoft.com/en-us/excel>

Common problem



Common Problems

Problem	What went wrong
Wrong value returned	If range_lookup is TRUE or left out, the first column needs to be sorted alphabetically or numerically. If the first column isn't sorted, the return value might be something you don't expect. Either sort the first column, or use FALSE for an exact match.
#N/A in cell	<ul style="list-style-type: none">▪ If range_lookup is TRUE, then if the value in the lookup_value is smaller than the smallest value in the first column of the table_array, you'll get the #N/A error value.▪ If range_lookup is FALSE, the #N/A error value indicates that the exact number isn't found.

Source: <https://support.microsoft.com/en-us/excel>

#REF! in cell	<p>If col_index_num is greater than the number of columns in table_array, you'll get the #REF! error value.</p> <p>For more information on resolving #REF! errors in VLOOKUP, see How to correct a #REF! error.</p>
#VALUE! in cell	<p>If the table_array is less than 1, you'll get the #VALUE! error value.</p> <p>For more information on resolving #VALUE! errors in VLOOKUP, see How to correct a #VALUE! error in the VLOOKUP function.</p>
#NAME? in cell	<p>The #NAME? error value usually means that the formula is missing quotes. To look up a person's name, make sure you use quotes around the name in the formula. For example, enter the name as "Fontana" in =VLOOKUP("Fontana",B2:E7,2,FALSE).</p> <p>For more information, see How to correct a #NAME! error.</p>
#SPILL! in cell	<p>This particular #SPILL! error usually means that your formula is relying on implicit intersection for the lookup value, and using an entire column as a reference. For example, =VLOOKUP(A:A,A:C,2,FALSE). You can resolve the issue by anchoring the lookup reference with the @ operator like this: =VLOOKUP(@A:A,A:C,2,FALSE). Alternatively, you can use the traditional VLOOKUP method and refer to a single cell instead of an entire column: =VLOOKUP(A2,A:C,2,FALSE).</p>

Source: <https://support.microsoft.com/en-us/excel>

Best practices



Best practices

Do this	Why
Use absolute references for <i>range_lookup</i>	<p>Using absolute references allows you to fill-down a formula so that it always looks at the same exact lookup range.</p> <p>Learn how to use absolute cell references.</p>
Don't store number or date values as text.	<p>When searching number or date values, be sure the data in the first column of <i>table_array</i> isn't stored as text values. Otherwise, VLOOKUP might return an incorrect or unexpected value.</p>
Sort the first column	<p>Sort the first column of the <i>table_array</i> before using VLOOKUP when <i>range_lookup</i> is TRUE.</p>

Source: <https://support.microsoft.com/en-us/excel>

Use wildcard characters

If ***range_lookup*** is FALSE and ***lookup_value*** is text, you can use the wildcard characters—the question mark (?) and asterisk (*)—in ***lookup_value***. A question mark matches any single character. An asterisk matches any sequence of characters. If you want to find an actual question mark or asterisk, type a tilde (~) in front of the character.

For example, =VLOOKUP("Fontan?",B2:E7,2,FALSE) will search for all instances of **Fontana** with a last letter that could vary.

Make sure your data doesn't contain erroneous characters.

When searching text values in the first column, make sure the data in the first column doesn't have leading spaces, trailing spaces, inconsistent use of straight (' or ") and curly (' or ") quotation marks, or nonprinting characters. In these cases, VLOOKUP might return an unexpected value.

To get accurate results, try using the [CLEAN function](#) or the [TRIM function](#) to remove trailing spaces after table values in a cell.

Thank you